

```

//*****
//*   MACRO: CHECKBOX.WCM
//*   PURPOSE: Creates checkboxes in the document that can be changed by clicking on
them.
//*****
Application (A1; "WordPerfect"; Default; "EN")
Constant (Checked = "☑")
Constant (Unchecked = "☐")
Constant (HypertextOffBegin = 14036)
Constant (HypertextOffEnd = 14292)

// Check to see if any text is selected. If it is, then move in front of it.
If (?BlockActive)
    PosBlockTop ()
    SelectOff ()
EndIf

Left := ?LeftChar
Right := ?RightChar
Switch (True)
    CaseOf Left = UnChecked:
        DeleteCharPrevious ()
        Type (Checked)
    CaseOf Left = Checked:
        DeleteCharPrevious ()
        Type (Unchecked)
    CaseOf Right = UnChecked:
        DeleteCharNext ()
        Type (Checked)
    CaseOf Right = Checked:
        DeleteCharNext ()
        Type (Unchecked)
    Default:
        If (?HypertextType = 0)
            Type (Unchecked)
            SelectCharPrevious ()
            HypertextCreate (RunMacro!; "checkboxbox.wcm"; ; HighlightedText!)
            PosCharNext ()
            If (?TextColorGreen)
                StyleEditBegin (HypertextStyle!; CurrentDoc!)
                StyleCodes (WithOffCodes!; CurrentDoc!)
                SelectAll ()
                DeleteCharNext ()
                SubstructureExit ()
                StyleEditEnd (Save!)
            EndIf
        EndIf

```

```
        If (?RightChar = UnChecked)
            PosCharNext ()
        EndIf
        If (Not ?ActivateHypertext)
            Hypertext (Active!)
        EndIf
    EndIf
EndSwitch
// Put the cursor right after the HypertextOff code
If (?RightCode = HypertextOffBegin)
    PosCharNext ()
    If (?LeftCode != HypertextOffEnd)
        PosCharPrevious ()
    EndIf
EndIf
```